Stirling's Formula and Some Coin Tossing

Mth 232 Jan 15 2000
Bent E. Petersen
Filename: stirling_coin_toss.mws

Stirling's Formula

Let

```
> S:=n->sqrt(2*Pi*n)*n^n*exp(-n);
```

$$S := n \rightarrow \sqrt{2\,\pi\,n}\;n^n\,\mathbf{e}^{(-n)}$$

Then $S(n)$ is a pretty good approximation to $n!$. One can prove this and even estimate the error but we can get experimental evidence just by plotting the quotient $\frac{S(n)}{n!}$

```
> plot(S(n)/n!,n=1..100,thickness=2);
```



As you can see the quotient is pretty close to 1. We can therefore use $S(n)$ to obtain a rough estimate of the binomial coefficient $C(n, k)$

```
> B:=(n,k)->S(n)/(S(k)*S(n-k));
```

$$B := (n, k) \rightarrow \frac{S(n)}{S(k)\,S(n-k)}$$

We expect $B(n, k)$ to be a pretty good approximation to $C(n, k)$ as long as n, k, and $n-k$ are all fairly large. We can get a feeling for it by comparing $B(2\,m, m)$ and $C(2\,m, m)$ for large m. Note for $m = 100$

these numbers are about 10^{59}, so even if they are relatively close the difference may be a very large number. Thus we compare them by looking at the quotient $\frac{B(2m, m)}{C(2m, m)}$

```
> plot(B(2*m,m)/binomial(2*m,m),m=1..150,thickness=2);
```



We see for $100 \le m$ we have a fairly good approximation.

A Coin Toss Example

Suppose we toss a fair coin $2m$ times. We may view a record of the outcomes as a sequence of H and T's of length $2m$. Since the coin is fair each sequence has the same probability of occuring, namely $2^{(-2m)}$. The number of sequences with exactly k heads is given by the number of ways to select the k positions in the the sequence that contain H's from the total $2m$ positions, thus $C(m, k)$ sequences.

Now let's find the probability $p(m)$ of tossing exactly m heads in a sequence of $2m$ coin tosses. This probability is clearly $C(2m, m)\, 2^{(-2m)}$. Except for small m this number is very time-consuming to compute, so we will approximate it by

```
> p:=m->B(2*m,m)*2^(-2*m);
```

$$p := m \to B(2m, m)\, 2^{(-2m)}$$

Maple can simplify this expression a great deal

```
> p:=unapply(simplify(p(m)),m);
```

$$p := m \to \frac{1}{\sqrt{\pi}\sqrt{m}}$$

Suppose now you toss a coin $2\,m$ times and count the number of heads. Suppose you perform this experiment N times and that Q of your sequences contain exactly m heads. Then $\frac{Q}{N}$ is a statistical approximation of the probability $\mathrm{p}(m)$. You could then use the formula above to estimate π. There's is something mysterious about tossing a coin to estimate π, don't you think?

I hasten to add that this method of estimating π is not practical since m and N both have to be inconveniently large to obtain a good estimate. Of course, you can simulate the coin tossing with a computer, but then you have to worry about the quality of your random number generator. Why not try it?

For example, suppose $m = 100$, $N = 10000$ and $Q = 563$. Then

```
> evalf(solve(563/10000=1/(sqrt(x)*sqrt(100)),x));
```

$$3.154882654$$

This is a made-up example. You are unlikely to do that well with such small m and N.

Finally, to wrap up, let's observe how close an approximation $\mathrm{p}(m)$ is to the actual theoretical probability $\mathrm{q}(m)$, say for $m = 1$ to $m = 30$ (in this range $.1 \le \mathrm{q}(m)$ so we are not dealing with very small numbers).

```
> q:=m->binomial(2*m,m)*2^(-2*m);
```

$$q := m \rightarrow \mathrm{binomial}(2\,m, m)\,2^{(-2\,m)}$$

```
> plot(q(m)-p(m),m=2..30,thickness=2);
```



Not too shabby!